Classification of generously transitive groups A permutation group $G \lt S_n$ is called generously transitive, if for each $i,j$ there exists a permutation that interchanges them. Is there a reasonable classification of such (finite) groups?
 A: I believe the answer is No, there is no good classification of these things. It might be helpful for you to know that a "generously transitive permutation group" is the same as a 2-star transitive group. (See for instance the introduction to "The k-star Property for Permutation Groups" by Clough, Praeger, Schneider.) I can email you a copy of this paper if you need it.
A: Clearly every generously transitive permutation group $G < {\rm S}_n$ acts transitively on $\{1,\dots,n\}$. Therefore we can find all generously
transitive permutation groups of degree $n$ by searching through the
transitive groups of degree $n$. The latter are available in the
Transitive Permutation Groups Library of GAP, for all $n \leq 30$.
The following GAP function tests whether a given group is generously
transitive:
IsGenerouslyTransitive := function ( G )

  local  n, i, j;

  n := LargestMovedPoint(G);
  if not IsTransitive(G,[1..n]) then return false; fi;
  if Transitivity(G,[1..n]) >= 2 then return true; fi;
  for i in [1..n] do
    for j in [i+1..n] do
      if   RepresentativeAction(G,[i,j],[j,i],OnTuples) = fail
      then return false; fi;
    od;
  od;
  return true;
end;

Now let's check how many of the transitive groups of degree $\leq 20$
are even generously transitive:
gap> List([2..20],NrTransitiveGroups);                                         
[ 1, 2, 5, 5, 16, 7, 50, 34, 45, 8, 301, 9, 63, 104, 1954, 10, 983, 8, 1117 ]
gap> Sum(last);
4722
gap> List([2..20],n->Number(AllTransitiveGroups(DegreeAction,n),
>                           IsGenerouslyTransitive));
[ 1, 1, 4, 4, 11, 5, 37, 22, 39, 6, 191, 7, 47, 63, 1239, 9, 605, 5, 870 ]
gap> Sum(last);
3166

Hence about two thirds of the transitive groups of degree $\leq 20$
are even generously transitive.
-- Therefore, perhaps rather than asking for a classification of generously
transitive groups, one might ask for a classification of those transitive
permutation groups which are not generously transitive.
